
Electronic Supplementary Material 4: Scripts and tasks to perform the 
homogenized- and micro- FE analyses 
 

mFE analysis 

$!       
$!         _/_/_/  _/_/_/    _/         
$!          _/    _/    _/  _/           Image Processing Language   
$!         _/    _/_/_/    _/  
$!        _/    _/        _/             (c) Eindhoven Univ. of Technology 
$!     _/_/_/  _/        _/_/_/_/            Bert v. Rietbergen 
$!         
$! 
$! Script for XtremeCT2 micro-FE analysis. 
$! Requires SCANCO IPLFE v1.16 or higher to run 
$! 
$! Call as: @mFE.com disk2:[microct.data.00001234.00005678]c0001234.aim 
$! 
$! with the numbers 00001234 and 00005678 replaced by the correct sample number and  
$! measurement number respectively, and the filename c0001234 replaced by the correct file  
$! name. In case the data is stored at another disk also the disk specification should be replaced. 
$! 
$! This script uses the E5E6 cortical/cancellous compartment definitions  
$! Required input files: _SEG.AIM and .GOBJ 
$! 
$! ----------------------------------------------------------------------------- 
$! Definition of base filename and jobname: 
$  FILENAME     = F$PARSE("''P1'",,,"NAME") - "_SEG" 
$  DEV          = F$PARSE("''P1'",,,"DEVICE") 
$  DIR          = F$PARSE("''P1'",,,"DIRECTORY") 
$  FILEBASE     = DEV + DIR + FILENAME 
$  JOB_NAME     = FILEBASE + "_MFE" 
$! ----------------------------------------------------------------------------- 
$! Definition of symbols used in the script: 
$  IPL_SEGAIM   = FILEBASE + "_SEG.AIM"         ! input: segmented aim file 
$  IPL_FNAME0   = JOB_NAME + ".AIM"             ! output: FE aim file  
$  IPL_FNAME1   = JOB_NAME + "_RESULT.AIM"      ! output: FE result aim file 
$  IPL_GOBJ0    = FILEBASE + ".GOBJ"            ! input: periosteal contour 
$  IPL_PROBLEM  = 33                            ! problem_nr  
$  IPL_FEPAR_NO = 43                            ! variable_nr for post-proc. 
$  IPL_FEA0     = 10000                         ! Young's modulus trab. tissue 
$  IPL_FEA1     = 10000                         ! Young's modulus cort. tissue 
$  IPL_FEA2     = "-0.05"                       ! Critical_volume for strength  
$  IPL_FEA3     = "0.01"                        ! Critical_value   ,,     ,,     
$! 
$! ---- Check if the required input files exist 
$! 



$  IF (F$SEARCH(IPL_SEGAIM) .EQS. "") 
$  THEN 
$     WRITE SYS$OUTPUT " -----------------------------------------------------------------------------" 
$     WRITE SYS$OUTPUT " ERROR: cannot find file: ",IPL_SEGAIM 
$     WRITE SYS$OUTPUT " This task can only run after the standard patient evaluation is 
completed 
$     WRITE SYS$OUTPUT " -----------------------------------------------------------------------------" 
$     STOP 
$  ENDIF        
$  IF (F$SEARCH(IPL_GOBJ0) .EQS. "") 
$  THEN 
$     WRITE SYS$OUTPUT " -----------------------------------------------------------------------------" 
$     WRITE SYS$OUTPUT " ERROR: cannot find file: ",IPL_GOBJ0 
$     WRITE SYS$OUTPUT " This task can only run after the standard patient evaluation is 
completed 
$     WRITE SYS$OUTPUT " -----------------------------------------------------------------------------" 
$     STOP 
$  ENDIF         
$  ON ERROR THEN EXIT 
$! ----------------------------------------------------------------------------- 
$! First part of the IPL script: pre-processing 
$! ----------------------------------------------------------------------------- 
$  IPLFE_BATCH 
 
! ---- Read segmented aim generated by the default image processing. For XT2 
!      this file already has different values for the cort/trab compartments 
 
/read 
  -name                      comb 
  -filename                  "IPL_SEGAIM 
 
! ---- only include the regions indicated by the gobj 
 
/gobj_maskaimpeel_ow 
  -input_output              comb 
  -gobj_filename             "IPL_GOBJ0 
  -peel_iter                 0 
 
! ---- Delete unconnected parts 
 
/cl_rank_extract 
  -input                     comb 
  -output                    cl 
  -first_rank                1 
  -last_rank                 1 
  -connect_boundary          false 
  -value_in_range            1 
 



/multiply_volumes 
  -input1                    comb 
  -input2                    cl 
  -output                    out 
  -common_region_only        false 
  -multiply_zero_pixels      true 
 
/del comb 
/del cl 
 
! ---- Remove empty space around the bone, if any 
 
/bounding_box_cut 
  -input                     out 
  -output                  fem 
  -z_only                   false 
  -border                  0 0 0 
 
/del out 
 
/write 
  -name                      fem 
  -filename                  "IPL_FNAME0         ! (_MFE.AIM) 
  -compress_type             bin 
.. 
$! ----------------------------------------------------------------------------- 
$! Second part of the IPL script: the actual FE-analysis 
$! ----------------------------------------------------------------------------- 
$  IPLFE_BATCH 
 
/read 
  -name                      fem 
  -filename                  "IPL_FNAME0         ! (_MFE.AIM) 
 
/fe_solve3 
  -in                        fem 
  -fea_file_name             "JOB_NAME           ! (_MFE) 
  -problem_nr                "IPL_PROBLEM 
  -scale_factor              -0.01 
  -list_option               1 
  -tolerance_force           1.000000E-03 
  -tolerance_displ           1.000000E+00 
  -max_nr_iter               50000 
  -restart_option            0 
  -comp_val_mat_001          126 
  -Ymodulus_mat_001          "IPL_FEA0 
  -Poissonr_mat_001          3.000000E-01 
  -comp_val_mat_002          127 



  -Ymodulus_mat_002          "IPL_FEA1 
  -Poissonr_mat_002          3.000000E-01 
  -comp_val_mat_003          0 
  -Ymodulus_mat_003          1.000000E+04 
  -Poissonr_mat_003          3.000000E-01 
.. 
$! ----------------------------------------------------------------------------- 
$! Third part of the IPL script: post-processing 
$! ----------------------------------------------------------------------------- 
$  IPLFE_BATCH 
 
/db_scanco_activate 
  -write                     true 
 
! ---- use the voxgobj to get the correct threshold in the database 
 
/read 
  -name                      fem 
  -filename                  "IPL_FNAME0         ! (_MFE.AIM) 
 
/voxgobj_scanco_param 
  -input                     fem 
  -gobj_filename             "IPL_GOBJ0 
  -peel_iter                 -1 
  -region_number             0 
 
/del fem 
 
! ---- do the postprocessing and store results in the database 
 
/fe_post 
  -post_file_name            "JOB_NAME           ! (_MFE) 
  -output                    res 
  -variable_nr               "IPL_FEPAR_NO 
  -loadcase_nr               1 
  -list_option               1 
  -interpol_option           1 
  -averaging_option          0 
  -test_dir                  0 
  -critical_variable_nr      "IPL_FEPAR_NO       ! Not available in FE v01.* 
  -critical_volume           "IPL_FEA2 
  -critical_value            "IPL_FEA3 
 
/write 
  -name                      res 
  -filename                  "IPL_FNAME1         ! (_MFE_RESULT.AIM) 
.. 
$! 



hFE analysis 

 
$!       
$!         _/_/_/  _/_/_/    _/         
$!          _/    _/    _/  _/           Image Processing Language   
$!         _/    _/_/_/    _/  
$!        _/    _/        _/             (c) Eindhoven Univ. of Technology 
$!     _/_/_/  _/        _/_/_/_/            Bert v. Rietbergen 
$!         
$!         
$! Script for XtremeCT2 homogenized-FE analysis  
$! Requires SCANCO IPLFE v2.01 or higher to run 
$! 
$! Call as: @hFE.com disk2:[microct.data.00001234.00005678]c0001234.aim 
$! 
$! with the numbers 00001234 and 00005678 replaced by the correct sample number 
$! and measurement number respectively, and the filename c0001234 replaced by  
$! the correct file name. In case the data is stored at another disk also the  
$! disk specification should be replaced. 
$! 
$! This script uses the E5E6 cortical/cancellous compartment definitions  
$! Required input files: .AIM, _SEG.AIM, .GOBJ and _TRAB_MASK.GOBJ 
$! 
$! ----------------------------------------------------------------------------- 
$! Definition of base filename and jobname: 
$  FILENAME     = F$PARSE("''P1'",,,"NAME") - "_SEG" 
$  DEV          = F$PARSE("''P1'",,,"DEVICE") 
$  DIR          = F$PARSE("''P1'",,,"DIRECTORY") 
$  FILEBASE     = DEV + DIR + FILENAME 
$  JOB_NAME     = FILEBASE + "_HFE" 
$! 
$! Definition of symbols used in the script: 
$  IPL_AIM      = FILEBASE + ".AIM"               ! input: aim file 
$  IPL_SEGAIM   = FILEBASE + "_SEG.AIM"           ! input: segmented aim 
$  IPL_FNAME0   = JOB_NAME + ".INP"               ! output: FE input inp 
$  IPL_FNAME1   = JOB_NAME + "_FST_INC_RESULT.AIM"! output: first inc res. 
$  IPL_FNAME2   = JOB_NAME + "_LST_INC_RESULT.AIM"! output: last inc res. 
$  IPL_FNAME3   = JOB_NAME + "_BASE.INP"          ! temp: model input 
$  IPL_FNAME4   = JOB_NAME + "_TRAB.INP"          ! temp: model input 
$  IPL_GOBJ0    = FILEBASE + ".GOBJ"              ! input: perios. contour 
$  IPL_GOBJ1    = FILEBASE + "_TRAB_MASK.GOBJ"    ! input: endost. contour 
$  IPL_MISC1_0  = 28                              ! downscale factor 
$  IPL_MISC1_1  = 1                               ! meshing_option 
$  IPL_MISC1_2  = 1                               ! fabric_option 
$  IPL_PROBLEM  = 33                              ! problem_nr  
$  IPL_FEPAR_NO = 42                              ! var. for first inc. 
$  IPL_FEA0     = 120                             ! var. for last inc. 



$! 
$! ---- Check if the required input files exist 
$! 
$  IF (F$SEARCH(IPL_AIM) .EQS. "") 
$  THEN 
$     WRITE SYS$OUTPUT " -----------------------------------------------------------------------------" 
$     WRITE SYS$OUTPUT " ERROR: cannot find file: ",IPL_AIM 
$     WRITE SYS$OUTPUT " This task can only run after the standard patient evaluation is 
completed 
$     WRITE SYS$OUTPUT " -----------------------------------------------------------------------------" 
$     STOP 
$  ENDIF        
$  IF (F$SEARCH(IPL_SEGAIM) .EQS. "") 
$  THEN 
$     WRITE SYS$OUTPUT " -----------------------------------------------------------------------------" 
$     WRITE SYS$OUTPUT " ERROR: cannot find file: ",IPL_SEGAIM 
$     WRITE SYS$OUTPUT " This task can only run after the standard patient evaluation is 
completed 
$     WRITE SYS$OUTPUT " -----------------------------------------------------------------------------" 
$     STOP 
$  ENDIF        
$  IF (F$SEARCH(IPL_GOBJ0) .EQS. "") 
$  THEN 
$     WRITE SYS$OUTPUT " -----------------------------------------------------------------------------" 
$     WRITE SYS$OUTPUT " ERROR: cannot find file: ",IPL_GOBJ0 
$     WRITE SYS$OUTPUT " This task can only run after the standard patient evaluation is 
completed 
$     WRITE SYS$OUTPUT " -----------------------------------------------------------------------------" 
$     STOP 
$  ENDIF         
$  IF (F$SEARCH(IPL_GOBJ1) .EQS. "") 
$  THEN 
$     WRITE SYS$OUTPUT " -----------------------------------------------------------------------------" 
$     WRITE SYS$OUTPUT " ERROR: cannot find file: ",IPL_GOBJ1 
$     WRITE SYS$OUTPUT " This task can only run after the standard patient evaluation is 
completed 
$     WRITE SYS$OUTPUT " -----------------------------------------------------------------------------" 
$     STOP 
$  ENDIF         
$! 
$! ---- When using hex elements, use a low mask threshold such that any element  
$!      touching the mask is included. When using tet elements, threshold at 50%  
$!      to get the proper contours 
$! 
$  IF (IPL_MISC1_1 .EQ. 1)  
$  THEN 
$     SCALE_THRESHOLD == 10 
$  ELSE  



$     IF (IPL_MISC1_1 .EQ. 2)  
$     THEN 
$        SCALE_THRESHOLD == 500 
$     ELSE 
$        WRITE SYS$OUTPUT " ---------------------------------------------------" 
$        WRITE SYS$OUTPUT " Invalid meshing option: ", IPL_MISC1_1 
$        WRITE SYS$OUTPUT " Valid options are 1 and 2 only 
$        WRITE SYS$OUTPUT " ---------------------------------------------------" 
$        EXIT 2 
$     ENDIF 
$  ENDIF 
$  ON ERROR THEN EXIT 
$! ----------------------------------------------------------------------------- 
$! First part of the IPL script: pre-processing and homogenization 
$! ----------------------------------------------------------------------------- 
$  IPLFE_BATCH 
 
! ---- Read the original input files  
 
/read 
  -name                      org_in 
  -filename                  "IPL_AIM 
 
/read 
  -name                      seg_in 
  -filename                  "IPL_SEGAIM 
 
/gobj_to_aim 
  -gobj_filename             "IPL_GOBJ0 
  -output                    full_mask_in 
  -peel_iter                 0 
 
/gobj_to_aim 
  -gobj_filename             "IPL_GOBJ1 
  -output                    trab_mask_in 
  -peel_iter                 0 
 
! ---- Make a common 'canvas' to which all images will be pasted to ensure 
!      they are positioned in the proper way relative to each other when  
!      removing the pos later 
 
/threshold 
  -input                     org_in 
  -output                    tmp 
  -lower_in_perm_aut_al      999999 
  -upper_in_perm_aut_al      0 
  -value_in_range            0 
  -unit                      0 



 
! ---- Set a border of downscale_factor voxels around the canvas in x/y-dir 
!      This is needed later for better density interpolation when using tet elements 
 
/border_change 
  -input                     tmp 
  -output                    canvas 
  -border                    "IPL_MISC1_0 "IPL_MISC1_0 0 
 
/del tmp 
 
! ---- Now place all images in the canvas  
 
/concat 
  -input1                    canvas 
  -input2                    org_in 
  -output                    org_canv 
  -common_region_only        false 
  -add_not_overlay           false 
 
/del org_in 
 
/concat 
  -input1                    canvas 
  -input2                    seg_in 
  -output                    seg_canv 
  -common_region_only        false 
  -add_not_overlay           false 
 
/del seg_in 
 
/concat 
  -input1                    canvas 
  -input2                    full_mask_in 
  -output                    full_mask_canv 
  -common_region_only        false 
  -add_not_overlay           false 
 
/del full_mask_in 
 
/concat 
  -input1                    canvas 
  -input2                    trab_mask_in 
  -output                    trab_mask_canv 
  -common_region_only        false 
  -add_not_overlay           false 
 
/del trab_mask_in 



 
! ---- Now all images are pasted in the same canvas, crop all images to the volume of  
!      interest (voi) which is defined as the volume of seg with a border in x/y-dir  
!      of downscale_factor voxels.  
!      This will remove any slices potentially lost due to slice matching from all images. 
 
! ---- first get the voi 
 
/gobj_maskaimpeel_ow 
  -input_output              seg_canv 
  -gobj_filename             "IPL_GOBJ0 
  -peel_iter                 0 
 
/bounding_box_cut 
  -input                     seg_canv 
  -output                    voi 
  -z_only                    false 
  -border                    "IPL_MISC1_0 "IPL_MISC1_0 0                ! Create the voi border in x/y-dir only 
 
/set_value 
  -input                     voi 
  -value_object              0 
  -value_background          0 
 
! ---- then crop all images to the voi dim and pos 
 
/concat 
  -input1                    voi 
  -input2                    org_canv 
  -output                    org 
  -common_region_only        true 
  -add_not_overlay           false 
 
/del org_canv 
 
/concat 
  -input1                    voi 
  -input2                    seg_canv 
  -output                    seg 
  -common_region_only        true 
  -add_not_overlay           false 
 
/del seg_canv 
 
/concat 
  -input1                    voi 
  -input2                    full_mask_canv 
  -output                    full_mask 



  -common_region_only        true 
  -add_not_overlay           false 
 
/del full_mask_canv 
 
/concat 
  -input1                    voi 
  -input2                    trab_mask_canv 
  -output                    trab_mask 
  -common_region_only        true 
  -add_not_overlay           false 
 
/del trab_mask_canv 
/del voi 
 
/exa org geo 
/exa seg geo 
/exa trab_mask geo 
/exa full_mask geo 
 
! ---- Set the pos of all files to zero such that alignment of downscaled and  
!      high-resolution images is exact for any down_scale_factor.  
!      For example, if this is not done and the pos was 49 and the downscale  
!      factor 10, the downscaled image would have a pos of 4. The error in the  
!      pos thus would be 9 voxels in size. This would affect the calculation of  
!      the element density/fabric later. By setting the pos to zero, no such  
!      errors occur 
 
/header_geo_set 
  -input                     org 
  -off_new                   -1   -1   -1 
  -pos_new                    0    0    0 
  -el_size_mm_new            -1.0 -1.0 -1.0 
 
/header_geo_set 
  -input                     seg 
  -off_new                   -1   -1   -1 
  -pos_new                    0    0    0 
  -el_size_mm_new            -1.0 -1.0 -1.0 
 
/header_geo_set 
  -input                     full_mask 
  -off_new                   -1   -1   -1 
  -pos_new                    0    0    0 
  -el_size_mm_new            -1.0 -1.0 -1.0 
 
/header_geo_set 
  -input                     trab_mask 



  -off_new                   -1   -1   -1 
  -pos_new                    0    0    0 
  -el_size_mm_new            -1.0 -1.0 -1.0 
 
/exa trab_mask geo 
 
! ------ create the cortical and trabecular compartment masks  
 
/subtract_aims 
  -input1                    full_mask 
  -input2                    trab_mask 
  -output                    cort_mask 
 
/set_value 
  -input                     trab_mask 
  -value_object              1 
  -value_background          0 
 
/set_value 
  -input                     cort_mask 
  -value_object              2 
  -value_background          0 
 
! ---- Generate the mesh 
 
/fe_mesh 
  -in                        full_mask 
  -out_filename              "IPL_FNAME3         ! (_HFE_BASE.INP) 
  -output_format             abaqus 
  -down_scale_factor         "IPL_MISC1_0 
  -threshold_value           "SCALE_THRESHOLD 
  -threshold_unit            6 
  -max_nr_groups             0 
  -meshing_option            "IPL_MISC1_1 
  -element_order             1 
  -use_hex_as_tet            false 
  -oddcase_start             true 
  -optimize_bw               fasle 
  -max_aspect_ratio          1.000000 
  -min_side_length           0.100000 
  -global_pos_flag           true 
  -problem_nr                0 
  -max_nr_mat                0 
  -density_option            0 
  -ref_density               1.000000 
  -Emod_const_a              10000.000000 
  -Emod_const_b              0.000000 
  -Emod_const_p              1.000000 



  -Poissons_ratio            0.300000 
 
/del full_mask 
 
! ---- calculate element density and fabric for the trabecular elements 
 
/hfe_homogenize 
  -org                       org 
  -seg                       seg 
  -mask                      trab_mask 
  -mesh_infile_name          "IPL_FNAME3         ! (_HFE_BASE.INP) 
  -mesh_outfile_name         "IPL_FNAME4         ! (_HFE_TRAB.INP) 
  -VOI_radius                2.000000 
  -volfrac_option            2 
  -volfrac_ref_density       1200.000000 
  -volfrac_min               0.001 
  -fabric_option             "IPL_MISC1_2 
  -fabric_min_volume         8.000000 
  -fabric_min_volfrac        0.050000 
  -fabric_power_transform    1.0 
  -MIL_peel_iter             0 
  -MIL_ip_sigma              0.000000 
  -MIL_ip_support            0 
  -MIL_ip_threshold          50 
  -MIL_nr_ave_iter           2 
  -MIL_t_dir_radius          2 
  -MIL_epsilon               1.200000 
  -list_option               2 
 
/del trab_mask 
 
! ---- calculate element density but no fabric for the cortical elements 
!      Set a very small radius: will be set to element size by /hfe_homogenize 
 
/hfe_homogenize 
  -org                       org 
  -seg                       seg 
  -mask                      cort_mask 
  -mesh_infile_name          "IPL_FNAME4         ! (_HFE_TRAB.INP) 
  -mesh_outfile_name         "IPL_FNAME0         ! (_HFE.INP) 
  -VOI_radius                0.010000 
  -volfrac_option            2 
  -volfrac_ref_density       1200.000000 
  -volfrac_min               0.001 
  -fabric_option             0 
  -fabric_min_volume         8.000000 
  -fabric_min_volfrac        0.050000 
  -fabric_power_transform    1.0 



  -MIL_peel_iter             0 
  -MIL_ip_sigma              0.000000 
  -MIL_ip_support            0 
  -MIL_ip_threshold          50 
  -MIL_nr_ave_iter           2 
  -MIL_t_dir_radius          2 
  -MIL_epsilon               1.200000 
  -list_option               2 
 
/del org 
/del seg 
/del cort_mask 
.. 
$  DEL 'IPL_FNAME3';* 
$  DEL 'IPL_FNAME4';* 
$! 
$! ----------------------------------------------------------------------------- 
$! Second part of the IPL script: the actual FE-analysis 
$! ----------------------------------------------------------------------------- 
$! 
$  IPLFE_BATCH 
 
/hfe_solve 
  -in_file_name              "IPL_FNAME0         ! (_HFE.INP) 
  -job_file_name             "JOB_NAME           ! (_HFE_XT2_STD) 
  -Eparams                   19010. 7851 0.223 
  -Epowers                   2.00 0.96 1.00 
  -Sparams                   131. 166. 67.3 
  -Spowers                   1.82 0.79 1.00 
  -CHI                       0.23 0.26 
  -Dparams                   0.750 7.000 
  -Hparams                   100.000 0.020 
  -problem_nr                "IPL_PROBLEM 
  -scale_factor              -0.05000000 
  -nod_select_tol            0.010 
  -nr_steps                  100 
  -tolerance                 0.001 
  -max_nr_iter               20 
  -assembly_every_iters      3 
  -max_force_drop            0.020000 
  -list_option               1 
.. 
$! 
$! ----------------------------------------------------------------------------- 
$! Third part of the IPL script: additional post-processing 
$! ----------------------------------------------------------------------------- 
$! 
$  POST1: 



$  IF (IPL_FEPAR_NO.EQ.0) THEN GOTO POST2 
$  IPLFE_BATCH 
 
/db_scanco_activate 
  -write                     true 
 
! ---- use the voxgobj to get the correct threshold in the database 
 
/read 
  -name                      in 
  -filename                  "IPL_SEGAIM 
 
/voxgobj_scanco_param 
  -input                     in 
  -gobj_filename             "IPL_GOBJ0 
  -peel_iter                 -1 
  -region_number             0 
 
! ---- create first plot 
 
/hfe_post 
  -job_file_name             "JOB_NAME           ! (_HFE_XT2_STD) 
  -out                       res 
  -dim_out                   -1 -1 -1 
  -pos_out                   -1 -1 -1 
  -el_size_mm_out            -1 -1 -1 
  -variable_nr               "IPL_FEPAR_NO 
  -loadstep_nr               1 
  -interpol_option           2 
  -defplot_scaling           0.000000 
  -list_option               1 
 
/write 
  -name                      res 
  -filename                  "IPL_FNAME1         ! (_HFE_FST_INC_RESULTS.AIM) 
  -compress_type             bin 
.. 
$! 
$  POST2: 
$  IF (IPL_FEA0.EQ.0) THEN EXIT 
$  IPLFE_BATCH 
 
/db_scanco_activate 
  -write                     true 
 
! ---- create 2nd plot  
 
/hfe_post 



  -job_file_name             "JOB_NAME           ! (_HFE_XT2_STD) 
  -out                       res 
  -dim_out                   -1 -1 -1 
  -pos_out                   -1 -1 -1 
  -el_size_mm_out            -1 -1 -1 
  -variable_nr               "IPL_FEA0 
  -loadstep_nr               99999 
  -interpol_option           2 
  -defplot_scaling           0.000000 
  -list_option               1 
 
/write 
  -name                      res 
  -filename                  "IPL_FNAME2         ! (_HFE_LST_INC_RESULTS.AIM) 
  -compress_type             bin 
.. 
$! 
 


